
1 The raw data file contains some partial information for commissioner service after
12/31/2000, but should be considered incomplete. The STATA infile statements provided with
these data therefore drop such observations from the final output.

Introduction
Overview

These data are intended to authoritatively document the personal, career, and political attributes
of every commissioner and board member who has served on any of the fifteen federal
independent regulatory agencies listed below, from March 31, 1887 (first day of operation for
the Interstate Commerce Commission) to December 31, 2000 (last date of data collection).1  The
database documents 743 individuals, who have been appointed a total of 1132 times (appointees
serve fixed and limited terms of service, but reappointments are common).  This research is
supported by a grant from the National Science Foundation (SES 00-95962), which began July 1,
2001 and ended August 31, 2004.  For each appointment, the data indicates sequential factors,
such as date of the nomination or the expiration of the term for the seat, and personal factors,
such as the gender or party affiliation of the appointee.  These data are intended to be the
counterpart to the appellate and district court biographical databases (Zuk, Barrow & Gryski
[ICPSR 6796], Goldman, Gryski & Zuk, respectively).  A permanent web site provides access to
the data and infile statements for these data at www.gsu.edu/~wwwirc.  In order to increase its
utility for a wide variety of potential applications, indicators on a broad range of variables of
theoretical significance to bureaucracy scholars were coded.  In addition, the data are assembled
in such a way that the unit of analysis may be either cross-sectional (the appointment or the
commissioner), or longitudinal (at any arbitary aggregation of time), or panel data.

Agencies Covered in these Data

Civil Aeronautics Board (1940-1981)  
Consumer Product Safety Commission (1972-2000)  
Equal Employment Opportunity Commission (1965-2000)  
Federal Communications Commission (1934-2000)
   & Federal Radio Commission (1928-1934)
Federal Election Commission (1975-2000)  
Federal Energy Regulatory Commission (1977-2000)
   & Federal Power Commission (1930-1977)
Federal Reserve Board (1914-1935) [excluding ex-officio members]
   & Board of Governors of the Federal Reserve (1935-2000) [political appointees only]
Federal Trade Commission (1914-2000)  
Interstate Commerce Commission (1887-1995)  
National Labor Relations Board (1935-2000) 
National Transportation Safety Board (1967-2000) 

Nuclear Regulatory Commission (1974-2000)
     & Atomic Energy Commission (1954-1974) 
Securities and Exchange Commission (1935-2000) 

Unit of Analysis

The fundamental data structure is as follows: one observation appears for each appointment of
each individual to have ever served on the respective agencies.  Each such appointment features
a unique 9-digit identification code of the following sort:

eeoc00502
______________   __________   _______
agency          individual    appointment

The first four digits indicate the name of the agency in a straightforward fashion (see Appendix 3
for explication of the agency abbreviations.  The next three digits are individual-specific,
providing a unique number for each commissioner or board member, in order of service on the
agency.  For any pairs of individuals who began service on the same date, those with a more
distant term expiration are coded with lower numbers.  For example, six of the first seven
commissioners on the FCC began service July 11, 1934, but each to a different term.  Thus, the
commissioner appointed to the longest term, Eugene Sykes, is coded as fcc001.  A researcher
may extract the first seven digits, and aggregate across them to obtain data in which the
commissioner is the unit of analysis.

The last two digits indicate the sequence of appointment, beginning with 01, the first.  In
all cases, if a commissioner was given a recess appointment, and was subsequently nominated
and confirmed, the situation is treated as a single appointment.  Special codes allow the
researcher to identify such situations, and the dates allow the researcher to identify all the key
actions taken.  Occasionally, commissioners may serve under recess appointments without being
confirmed (see Fisher 1997, 4th ed.).  Such service is documented in this dataset, and special
codes allow the researcher to identify such situations.  More frequently, persons are considered
as potential nominees, and sometimes actually nominated, but never confirmed.  In the absence
of a recess commission, they never actually serve on the agency, and these individuals are not
documented in this data set.

Variables

Variables in this data set may be categories in one of three groups:
1. Identification codes.  A number of codes have been assembled to allow researchers to identify
each unique person or appointment denoted in this data set.

a. appointment ID

b. appointment ID of immediate predecessor

2. Biographical.  Time invariant characteristics of individuals are provided, including party ID,
state of residence at time of appointment, gender, race/ethnicity, date of birth, and prior and
subsequent employment experience.  The coding scheme for employment experience is rooted in
a number of  “revolving door” analyses published in the literature (Gormley 1979; Eckert 1981;
Cohen 1986; Spiller 1990), but go beyond just conflict of interest indicators to provide data on
academic employment, public sector employment, and elective office.  Data on occupational
field are not explicitly provided because there is seldom any significant variation for
commissioners on a given agency.

a. state of residence (at time of appointment)
b. party affiliation
c. gender
d. race/ethnicity
e. birthdate
f. name
g. employment prior to agency service
h. employment following agency service

3. Appointment Circumstances.  For each appointment, the relevant circumstances have been
denoted, including the ID of the immediate predecessor, the dates of nomination, confirmation,
beginning of service, end of service, and term expiration.  Dates of Senate committee hearings
have not been denoted in this data set.

a. reason for termination of service under this appointment
b. temporary service (holdover or recess)
c. date of nomination
d. date of confirmation
e. first date of service under this appointment
f. last date of service under this appointment
g. expiration date of term for this appointment

Additional Materials

A large appendix of subjective coding decisions and various data idiosyncracies is
available from the Principal Investigator, upon request.  Such information includes, for example:
a list of bank board memberships held concurrently to the principal pre-appointment
employment for Fed board members; notation of the nomination withdrawl dates for recess
appointees never confirmed; identification of the US Code citation and effective date for the
public law reducing or expanding the number of seats on a commission; and so on.

In addition, the primary source materials for the entire project (approximately 12 linear
shelf-feet) is archived both in paper and electronically, and copies are available upon request
from the Principal Investigator. These files are constantly being updated, as new information
arises. Scholars who happen upon data discrepancies or holes are encouraged to pursue them
with the PI, for the benefit of the entire research community.

VARIABLE DESCRIPTION
A. IDENTIFIERS

Field 1
APPID

9 columns wide (1-9)
alphanumeric

This field represents a simple unique identifier for each appointment.  The first four
digits indicate the agency (see Appendix 3).  The next three digits are specific to the individual
commissioner, by agency, and indicate the consecutive order in which the commissioners served. 
In all cases where more than one commissioner began service on the same date, the
commissioner appointed to the longer term of service is given a lower number.  The last two
digits indicate, for each commissioner, the sequence of appointments, ranging from 1 to 4 (first
to fourth appointment, for a given commissioner).

Field 2
PREDID

9 columns wide (11-19)
alphanumeric

This field provides the ID for the immediate predecessor to the appointment.  In the case
of reappointments, a commissioner is her own predecessor, from the standpoint of this field. 
When no predecessor exists (ie for the first appointment to each seat after an agency is newly
created), the field is coded as “xxxxxxxxx”.

B. BIOGRAPHICAL
Field 3

STATE
2 columns wide (21-22)
alphanumeric

This field indicates the state of residence for the appointee at the time of his/her first
appointment.  Values follow the traditional 2-digit postal coding for states.  In situations where
residence is ambiguous or a commissioner plausibly has more than one residence, the
introduction of the nominee in the Senate (which invariably indicates the home state of the
nominee) is taken as definitive.  Thus, this field can be used to ascertain the Senators for whom
any courtesy might be extended.

Field 4
PARTY

1 column wide (24)
alphanumeric

This field indicates the party affiliation of the nominee at the time of first appointment. 
This is the definitive information used for ascertaining statutory compliance with party affiliation
limits in charter statutes.

Valid values: R - Republican
D - Democrat
I - Independent or Other

Field 5
GENDER

1 column wide (26)
alphanumeric

Valid values: M - Male
F - Female

Field 6
RACE

1 column wide (28)
numeric

This field categorizes appointees into one of four races/ethnicities: white, black, hispanic,
and asian.

Valid Values: 0 - white/caucasian
1 - african american
2 - hispanic/latino
3 - asian

Field 7
BIRTH

10 columns wide (30-39)
date-time (mm_dd_yyyy)

Appointee’s birthdate, ranging from 06jan1824 to 23mar1963.  For some individuals, the
actual day or even month may be missing (blank).

Field 8
NAME

25 columns wide (41-65)
string

Name is presented in the manner it was normally printed by the commission (sometimes
with middle initial, sometimes without).  This field contains embedded blanks and punctuation. 
Complete order of names is as follows: [last][,][first][middle][maiden][jr./III,etc].  In situations
where a name changed during service on the commission (marraige), the most complete name is

presented.
Fields 9-12

PREEMP1 – PREEMP4

each 2 columns wide (67-68, 70-71, 73-74, 76-77)
numeric

Codes for employment prior to (pre-) and after (post-) service on commission utilize the
same typology.  Codes are rooted in previous analyses of revolving door and career paths of
regulatory commissioners (Gormley 1979; Eckert 1981; Cohen 1986; Spiller 1990).  This dataset
provides indicators for the four most recent employment experiences of the commissioner, prior
to service (as well as the first four subsequent employment experiences, after service on the
commission).  Thus, the codes could be used to construct alternative employment measures, such
as a dummy variable for prior academic employment, by logically searching across the prior
employment fields.  For example: PREACADM=1 if (PREEMP1=8 or PREEMP2=8 or
PREEM3=8 or PREEMP4=8).  Variables are recorded chronologically, but variable labels are
presented in reverse order, for this variable.  Thus, PREEMP1 appears in columns 76-77, and
indicates the most recent employment experience prior to service on the commission.

Valid Values
1 - direct employment by regulated industry
2 - indirect employment (consulting, legal representation) by regulated industry
3 - public interest employment (typically non-profit consulting/executive positions)
4 - commission staff
5 - other federal agency staff
6 - state agency staff
7 - congressional staff
8 - academia/think tank
9 - unrelated private sector
10 - unrelated public sector
11 - state elective office
12 - federal elective office
13 - appointive state commissioner/agency head
14 - appointive federal commissioner/agency head
15 - state judicial
16 - federal judicial
17 - farmer
18 - related military service
19 - related public sector
20 - ambassadorial and international commission

Fields 13-16
POSTEMP1 – POSTEMP4

each 2 columns wide (79-80, 82-83, 85-86, 88-89)
numeric

Same coding as pre-employment, with POSTEMP1 indicating the most immediate
employment after service on commisson.

C. APPOINTMENT CIRCUMSTANCES
Field 17

EXIT
1 column wide (91)
numeric

Reason for termination of service.

Valid Values 1 - died in office
2 - resigned before term expired
3 - term expired, without renomination
4 - term expired, renominated
5 - never confirmed/nomination withdrawn
6 - transferred to new seat
7 - impeached
8 - still serving as of 12/31/2000
9 - seat abolished

Field 18
HOLDOVER

1 column wide (93)
numeric

For most agencies since about 1960, commissioners are permitted to continue serving on
the agency after their term has expired, until a replacement has been confirmed.  This dummy
(0=no holdover service, 1=holdover) indicates whether the commissioner served past the
expiration of the current term in this fashion. In such cases, end of service date will extend past
the expiration date.

Field 19
RECESS

1 column wide (95)
numeric

Under Article II of the U.S. Constitution, the President may grant a recess appointment
while the Senate is in recess.  Said appointments expire at the end of the following session of
Congress.  In addition, an 1863 law stipulates that recess appointees receive no financial
compensation for their service, unless they are subsequently confirmed by the Senate (Fisher
1997).  As described in the “Unit of Analysis” section of this codebook, recess appointments
subsequently confirmed are treated as a single observation.  The first date of service indicates the

date of the recess commission.  To guarantee that all commissioners who ever voted on a case
are documented in this dataset, recess appointees who serve but are never confirmed also are
treated as observations, and the variable EXIT indicates such situations.

Valid values 0 - normal appointment
1 - began service under a recess appointment

Field 20
NOM

10 columns wide (97-106)
date-time (mm_dd_yyyy)

Date of first nomination.  Subsequent nominations (such as at the beginning of a session
after the previous nomination expired at the end of the last session of Congress) are not
documented in this data set.

Field 21
CONF

10 columns wide (108-117)
date-time (mm_dd_yyyy)

Date of confirmation.
Field 22

BEG
10 columns wide (119-128)
date-time (mm_dd_yyyy)

First date of service.  For new commissioners, this is the date the oath of office was
administered, whether they received a normal or recess commission.  For reappointed
commissioners, the oath of office may not be immediately administered at the beginning of their
new term yet the commissioners are immediately serving (under holdover provision).  In those
cases, first date of service is the first day of their new term.  This coding convention allows one
to accurately reconstruct the continuous service of reappointed commissioners.

Field 23
END 

10 columns wide (130-139)
date-time (mm_dd_yyyy)

Last date of service.  If this date exceeds the expiration of the term, the situation is
indicated elsewhere in the data (HOLDOVER=1).

Field 24
EXP

10 columns wide (141-150)
date-time (mm_dd_yyyy)

Date term is statutorily set to expire.

DATA INFILE STATEMENTS
Data for each agency is appended into a single file, IRC.RAW. Subsetting the analysis

for a single agency is trivial, and the infile statements provided extract data for the EEOC, by
default. The infile statement provided, IRC.DO, contains multiple modules that may be
commented out, depending on your needs. By default, the infile statements execute each module
for EEOC appointments, producing 4 datasets at different levels of aggregation, as detailed
below.

Cross-Sectional Designs

In the raw form for distribution, the unit of observation for the data is the appointment.  Thus, no
aggregation is necessary if a researcher wishes to examine each appointment episode.  The
following infile statements (the first module of IRC.DO) read the data, and assemble date
variables in date-time format:

********************************************************************************
**************************** MODULE 1 ******************************************
*********** THIS READS IN RAW DATA AND PUTS OUT INTO STATA FORMAT **************
infix str agency 1-4 str id 1-7 sequence 8-9 str appid 1-9 str predid 11-19  ///
 str state 21-22 str party 24 str sex 26 race 28 str bday 30-39              ///
 str name 41-65 preemp4 67-68 preemp3 70-71 preemp2 73-74 preemp1 76-77      ///
 postemp1 79-80 postemp2 82-83 postemp3 85-86 postemp4 88-89                 ///
 exit 91 holdover 93 recess 95 str nom 97-106 str conf 108-117               ///
 str start 119-128 str end 130-139 str expire 141-150                        ///
 using "d:\projects\NSF\data\ICPSR\irc.raw"
drop if agency!="eeoc"                               // REPLACE WITH YOUR AGENCY
gen birthday=date(bday,"mdy")
gen nominated=date(nom,"mdy")
gen confirmed=date(conf,"mdy")
gen started=date(start,"mdy")
gen ended=date(end,"mdy")
gen expired=date(expire,"mdy")
format birthday nominated confirmed started ended expired %d
replace ended=d(31dec2000) if(ended==. | ended>d(31dec2000))  // DATA AFTER 31dec2000 IS
replace exit=8 if ended==d(31dec2000)                         //    IS
drop if started>d(31dec2000)                                  //  PARTIAL
label var birthday "date of birth"
label var nominated "date nominated for this appid"
label var confirmed "date confirmed for this appid"
label var started "first date of service under this appid"
label var ended "last date of service under this appid"
label var expired "term expiration date for this appid"
gen staffexp=(preemp1==4|preemp2==4|preemp3==4|preemp4==4)
label var staffexp "prior employmt as staff in own agency"
drop bday name nom conf start end expire preemp1 preemp2 preemp3 preemp4
gen cparty=party=="R"
replace cparty=-1 if party=="D"
replace cparty=0 if party=="I"
label define partyidvalues -1 "D" 0 "I" 1 "R"
label values cparty partyidvalues
label var cparty "commissioner partyID (-1,0,1)"
gen gender=sex=="F"
label define gendervalues 0 "M" 1 "F"
label values gender gendervalues
label var gender "gender (0-M 1-F)"
label var race "race (1-white, 2-afram, 3-hisp, 4-ais)"

gen served=ended-started+1
sort appid
save "d:\projects\NSF\data\ICPSR\ircbase.dta", replace

Within the same framework (one observation per appointment episode), some merging of
the data is possible, so as to append any or all characteristics of the immediate predecessor.  For
example this capability is helpful for analysis of the length of time a vacancy persisted on the
commission (e.g. Nixon 2001; Nixon & Goss 2001; Nixon & Bentley 2005).  A second portion
of the infile statement accomplishes this task:

********************************************************************************
**************************** MODULE 1b *****************************************
**** THIS LINKS TO PREDECESSORS, AND PUTS OUT A DATASET OF VACANCIES ***********
gen predexit=exit
gen predhold=hold
gen predend=ended
gen predparty=cparty
gen predage=(ended-birthday)/365.25
gen predgend=gender
gen predrace=race
replace predend=expired if hold==1                   // REMOVES HOLDOVER SERVICE
replace predid=appid                                // BOOKKEEPING FOR THE MERGE
drop if exit==8             // DROPS SERVING COMMISSIONERS FROM PREDECESSOR DATA
drop if exit==5                 // DROPS RECESS APPOINTEES FROM PREDECESSOR DATA
keep predid predexit predhold predparty predage predgend predrace predend
sort predid
save "d:\projects\NSF\data\ICPSR\ircpred.dta", replace
clear
use "d:\projects\NSF\data\ICPSR\ircbase.dta"
********************************************************************************
******* THIS SECTION SLICES OUT UNCONFIRMED RECESS APPOINTEES ******************
*******        AND STICHES TOGETHER CONFIRMED APPOINTEES      ******************
*******        SEE NIXON 2001; NIXON & BENTLEY 2004           ******************
drop if exit==5
replace predid="cpsc00401" if appid=="cpsc00502"
replace predid="eeoc02001" if appid=="eeoc02502"
replace predid="xxxxxxxxx" if appid=="fcc00801"
replace predid="fcc02702" if appid=="fcc03401"
replace predid="fec00801" if appid=="fec00202"
replace predid="fed00701" if appid=="fed01001"
replace predid="ftc01001" if appid=="ftc01501"
replace predid="ftc01601" if appid=="ftc01901"
replace predid="ftc03601" if appid=="ftc04001"
replace predid="ftc05702" if appid=="ftc06601"
replace predid="icc01002" if appid=="icc01401"
replace predid="icc01702" if appid=="icc03101"
replace predid="icc02901" if appid=="icc03801"
replace predid="icc06204" if appid=="icc08201"
replace predid="icc09901" if appid=="icc10201"
replace predid="nlrb00401" if appid=="nlrb00501"
replace predid="nlrb01901" if appid=="nlrb02301"
replace predid="nlrb03201" if appid=="nlrb03401"
replace predid="nlrb04201" if appid=="nlrb04901"
replace predid="nlrb04402" if appid=="nlrb05001"
replace predid="nlrb02902" if appid=="nlrb03801"
replace predid="nlrb02105" if appid=="nlrb03701"
replace predid="nlrb03901" if appid=="nlrb04501"
replace predid="nlrb03701" if appid=="nlrb04601"
replace predid="nlrb04901" if appid=="nlrb05201"
replace predid="nlrb04601" if appid=="nlrb04801"
replace predid="nlrb05001" if appid=="nlrb05401"
replace predid="nlrb05201" if appid=="nlrb05303"
replace predid="ntsb00104" if appid=="ntsb01901"
replace predid="sec03301" if appid=="sec03801"
replace predid="sec06402" if appid=="sec06802"
******* THIS SECTION SLICES OUT UNCONFIRMED RECESS APPOINTEES ******************
*******        AND STICHES TOGETHER CONFIRMED APPOINTEES      ******************
*******        SEE NIXON 2001; NIXON & BENTLEY 2004           ******************

********************************************************************************
drop if predid=="xxxxxxxxx"                        // DROPS INITIAL APPOINTMENTS
sort predid
merge predid using "d:\projects\NSF\data\ICPSR\ircpred.dta"
drop if _merge==2
drop _merge
gen renom=predexit==4
gen died=predexit==1
gen vacancy=started-predend                                 // APPOINTMENT DELAY
replace vacancy=confirmed-predend if recess==1
save "d:\projects\NSF\data\ICPSR\ircvac.dta", replace

An alternative cross-sectional design is possible, whereby the individual commissioner
serves as the unit of analysis. Such a design is trivially easy to construct, by dropping all those
observations for which SEQUENCE is greater than 1. The third module of IRC.DO generates
such a cross-sectional design, in a slightly more complicated way that preserves information
about later appointments.

********************************************************************************
**************************** MODULE 1c *****************************************
******** THIS COLLAPSES ALL APPOINTMENTS INTO A BIOGRAPHICAL DATASET *******
use "d:\projects\NSF\data\ICPSR\ircbase.dta", clear
sort id
collapse (min) fstarted=started fconfirmed=confirmed  ///
 cparty gender race=race birthday=birthday ///
 (max) lended=ended (count) totapps=sequence, by(id)
format %d fstarted fconfirmed lended
label var fconfirmed "very first confirmation date"
label var fstarted "very first date of service"
label var lended "very last day of service"
label var gender "gender (0-M 1-F)"
label var race “race (0-white 1-af/am 2-hisp 3-aisan)"
label var cparty "commissioner's party (-1=D 0=I +1=R)"
label var birthday "birthday"
label var totapps "total number of appointments"
sort id
save "d:\projects\NSF\data\ICPSR\ircbio.dta", replace     

Panel Designs

Researchers may prefer to aggregate the data into panel designs.  Probably the easiest
way to accomplish this task is to make sure two fields are available for each record: the date the
commissioner began service, and the number of days that service persisted.  If these two
variables are available, one may use a functionality new with version 7.0 of Stata, STSPLIT, to
spread the existing data into a panel design.  The following .DO commands generate a panel
design of one observation for each commissioner for each day they served:

********************************************************************************
**************************** MODULE 2 ******************************************
******** THIS SPLITS SERVICE INTO DAILY SPELLS *********************************
use "d:\projects\NSF\data\ICPSR\ircbase.dta", clear
drop nominated confirmed expired sequence state party
sort id
merge id using "d:\projects\NSF\data\ICPSR\ircbio.dta"
gen fail=1
stset served, id(appid) failure(fail)
stsplit stemp, every(1)
gen date=started+stemp
format date %d
drop if date>d(31dec2000)               // DATA PROVIDED AFTER 31dec ARE PARTIAL
gen age=(date-birthday)/365.25



Page 13

gen seniority=(date-fstarted+1)/365.25
label var seniority "years served since first appointed"
********************************************************************************
******* THIS SECTION ADJUSTS SENIORITY FOR THOSE *******************************
*******  COMMISSIONERS WITH GAPS IN THEIR SERVICE ******************************
replace seniority=(date-fstarted+1-5093)/365.25 if appid=="ftc05602" /* Pitofsky gap 1981-95   */
replace seniority=(date-fstarted+1-4384)/365.25 if appid=="fcc00902" /* Craven gap 1944-56     */
replace seniority=(date-fstarted+1-1178)/365.25 if appid=="aec00102" /* Strauss gap 1950-53    */
replace seniority=(date-fstarted+1-2)/365.25 if appid=="cab00602"    /* Warner gap 2 days 1941 */
replace seniority=(date-fstarted+1-565)/365.25 if appid=="fec00402"  /* Thomson gap 1979-81    */
replace seniority=(date-fstarted+1-559)/365.25 if appid=="icc09302"  /* Simmons gap 1983-84    */
replace seniority=(date-fstarted+1-559)/365.25 if appid=="icc09303"
replace seniority=(date-fstarted+1-559)/365.25 if appid=="icc09304"
******* THIS SECTION ADJUSTS SENIORITY FOR THOSE *******************************
*******  COMMISSIONERS WITH GAPS IN THEIR SERVICE ******************************
********************************************************************************
drop served fail _st _d _t0 stemp
save "d:\projects\NSF\data\ICPSR\ircsplit.dta", replace 

However, a researcher seeking to establish a monthly panel design probably prefers that the
“strobe” be the same day for each commissioner, which the above does not accomplish. It is best
to generate the daily panel and then drop all observations other than the first day of each month,
as in DROP IF DOM(DATE)!=1.

Longitudinal Designs

Many classical studies of bureaucracy and bureaucratic control are longitudinal in nature. 
One may easily aggregate across a variety of commissioner indicators, for purposes of time
series analysis.  For example, the final module of the infile statements generate a daily time
series containing (among other things), the proportion of sitting commissioners who are
democrats.  Variations on this theme are trivially easy to incorporate into a researcher’s version
of the infile statements.

********************************************************************************
**************************** MODULE 3 ******************************************
*********** THIS CREATES TIME SERIES INDICATORS FOR THE COMMISSION *************
************ IT IS EASILY EXTENSIBLE TO GENERATE ADDITIONAL INDICATORS *********
***********     IRCSERIES is daily, IRC_SMALLSERIES is strobed at the first of each month
********
sort agency date
gen repub=cparty==1
gen dem=cparty==-1
collapse (mean) majparty=cparty experience=seniority wisdom=age female=gender ///
(count) serving=cparty (sum) rserving=repub dserving=dem, by(agency date)
label var serving "number of commissioners serving"
label var rserving "number of Republicans serving"
label var serving "number of Democrats serving"
label var majparty "partisan balance (-1,0,1)"
label var experience "average years of prior service on the commission"
label var wisdom "average age of commissioners"
label var female "proportion of sitting commissioners who are female"
save "d:\projects\NSF\data\ICPSR\ircseries.dta", replace
drop if (day(date))!=1
gen date2=(year(date)) + ( (doy(date)) - 1)/365.25
label var date2 "year and month (decimalized)"

save "d:\projects\NSF\data\ICPSR\irc_smallseries.dta", replace     

APPENDIX 1
VARIABLE LIST IN ORDER OF APPEARANCE IN
DATA FILE
NAME
APPID
PREDID
STATE
PARTY
GENDER
RACE
BIRTH
NAME
PREEMP4
PREEMP3
PREEMP2
PREEMP1
POSTEMP1
POSTEMP2
POSTEMP3
POSTEMP4
EXIT
HOLDOVER
RECESS
NOM
CONF
BEG
END
EXP

BRIEF DESCRIPTION
appointment ID
predecessor’s appointment ID
residence at first appointment
party affiliation
gender
race/ethnicity
birthdate
name
fourth most recent employment prior to service
third-most recent prior employment
second-most recent prior employment
most recent employment prior to service
first employment following service
second employment after service
third employment after service
fourth employment after service
reason for service termination
dummy for holdover service
dummy for recess commission
date nominated
date confirmed
date began service
date ended service
term expiration date

FORMAT
char9.
char9.
char2.
char1.
char1.
numeric
mmddyyyy
char25.
numeric
numeric
numeric
numeric
numeric
numeric
numeric
numeric
numeric
numeric
numeric
mmddyyyy
mmddyyyy
mmddyyyy
mmddyyyy
mmddyyyy

LOCATION
1-9

11-19
21-22

24
26
28

30-39
41-65
67-68
70-71
73-74
76-77
79-80
82-83
85-86
88-89

91
93
95

97-106
108-117
119-128
130-139
141-150



Page 15

APPENDIX 2
VARIABLE LIST IN ALPHABETICAL ORDER
NAME
APPID
BEG
BIRTH
CONF
END
EXIT
EXP
GENDER
HOLDOVER
NAME
NOM
PARTY
POSTEMP1
POSTEMP2
POSTEMP3
POSTEMP4
PREDID
PREEMP1
PREEMP2
PREEMP3
PREEMP4
RACE
RECESS
STATE

BRIEF DESCRIPTION
appointment ID
date began service
birthdate
date confirmed
date ended service
reason for service termination
term expiration date
gender
dummy for holdover service
name
date nominated
partisan affiliation
first employment following service
second employment after service
third employment after service
fourth employment after service
predecessor’s appointment ID
most recent employment prior to service
second-most recent prior employment
third-most recent prior employment
fourth-most recent prior employment
race/ethnicity
dummy for recess commission
residence at first appointment

FORMAT
char9.
mmddyyyy
mmddyyyy
mmddyyyy
mmddyyyy
numeric
mmddyyyy
char2.
numeric
char25.
mmddyyyy
char1.
numeric
numeric
numeric
numeric
char9.
numeric
numeric
numeric
numeric
numeric
numeric
char2.

LOCATION
1-9

119-128
30-39

108-117
130-139

91
141-150

26
93

41-65
97-106

24
79-80
82-83
85-86
88-89
11-19
76-77
73-74
70-71
67-68

28
95

21-22

APPENDIX 3
List of Agency Codes/Abbreviations
AEC Atomic Energy Commission (1954-1974) 
BGFR Board of Governors of the Federal Reserve (1935-2000)
CAB Civil Aeronautics Board (1940-1981)  
CPSC Consumer Product Safety Commission (1972-2000)  
EEOC Equal Employment Opportunity Commission (1965-2000)  
FCC Federal Communications Commission (1934-2000)
FEC Federal Election Commission (1975-2000)  
FED Federal Reserve Board (1914-1935)
FERC Federal Energy Regulatory Commission (1977-2000)
FPC Federal Power Commission (1930-1977)  
FRC Federal Radio Commission (1928-1934)
FTC Federal Trade Commission (1914-2000)  
ICC Interstate Commerce Commission (1887-1995)  
NLRB National Labor Relations Board (1935-2000) 
NRC Nuclear Regulatory Commission (1974-2000)
NTSB National Transportation Safety Board (1967-2000) 
SEC Securities and Exchange Commission (1935-2000)


